Notes on Chapter 11 of Special Functions

George Beck, July 8, 2020

11.1 Background on Partitions

A partition of a positive integer *n* is a multiset of positive integers that add up to *n*. (A multiset is like a set in that order does not matter but unlike a set in that elements can be repeated.) For example, these are the seven partitions of five:

In[•]:=
```
IntegerPartitions@4
```
Out[•]=
```
{{4}, {3, 1}, {2, 2}, {2, 1, 1}, {1, 1, 1, 1}}
```

These can be written symbolically (because the addition is frozen):

In[•]:=
```
Row[#, "+"] & /@ IntegerPartitions@4
```
Out[•]=
```
{4, 3 + 1, 2 + 2, 2 + 1 + 1, 1 + 1 + 1 + 1}
```

$p(n)$

The partition function $p(n)$ counts the number of partitions of *n*.

In particular, since there are five partitions of 4, $p(4) = 5$.

Here are the first 10 values of $p(n)$ (A000041):

In[•]:=
```
PartitionsP@Range@10
```
Out[•]=
```
{1, 2, 3, 5, 7, 11, 15, 22, 30, 42}
```

The function $p(n)$ grows very quickly; $p(200)$ is nearly 4 trillion.

In[•]:=
```
PartitionsP@200
```
Out[•]=
```
3972999029388
```

$p_m(n)$

The function $p_m(n)$ counts the number of partitions of *n* into *m* or fewer parts.

Euler's table of $p_m(n)$ (A026820) has *n* increasing down and *m* across.

In[∘]:=
```
Table[Length@IntegerPartitions[n, m], {n, 10}, {m, n}] // Grid
```

Out[∘]=

1									
1	2								
1	2	3							
1	3	4	5						
1	3	5	6	7					
1	4	7	9	10	11				
1	4	8	11	13	14	15			
1	5	10	15	18	20	21	22		
1	5	12	18	23	26	28	29	30	
1	6	14	23	30	35	38	40	41	42

The main diagonal is $p(n)$, because if n or fewer parts are allowed, there is no restriction.

The first column is all 1's because there is only one partition of n with one part, namely n.

For the second column $p_2(n)$, count the partitions into two parts $\{n-1, 1\}$, $\{n-2, 2\}$, …, $\{\lceil n/2 \rceil, \lfloor n/2 \rfloor\}$ and the one partition in one part n for a total of $\lceil n/2 \rceil + 1$. In Mathematica the ceiling function $\lceil \cdot \rceil$ is Ceiling.

In[∘]:=
```
Table[Ceiling[n/2] + 1, {n, 2, 10}]
```

Out[∘]= {2, 3, 3, 4, 4, 5, 5, 6, 6}

$p_{\mathbb{N}}^{(1)}(n)$

A distinct partition has no repeated part. In other words, the multiset of parts is a set.

For example, here are the two distinct partitions of four:

In[∘]:=
```
Select[IntegerPartitions[4], Length@# == Length@Union@# &]
```

Out[∘]= {{4}, {3, 1}}

The function $p_{\mathbb{N}}^{(1)}(n)$ counts the number of distinct partitions of n.

(In the literature, this is sometimes denoted $q(n)$, but q is used in the book for q-series.)

The sequence $p_{\mathbb{N}}^{(1)}(n)$ (A000009) grows more slowly than $p(n)$.

These are the first 10 values of $p_{\mathbb{N}}^{(1)}(n)$:

In[∘]:=
```
PartitionsQ@Range@10
```

Out[∘]= {1, 1, 2, 2, 3, 4, 5, 6, 8, 10}

Compared to $p(200) \approx 4$ trillion, $p_{\mathbb{N}}^{(1)}(200)$ is less than half a billion:

```
PartitionsQ@200
```

Out[∘]= 487067746

$p_A(n)$

Let $p_A(n)$ be the number of partitions of n into elements of $A \subseteq \mathbb{N}$.

For example, let $\mathcal{B} = \{1, 2, 4, 8, \ldots\}$. These are the binary partitions of 12 written compactly:

```
In[•]:= Row /@ (bin12 = IntegerPartitions[12, All, {1, 2, 4, 8}])
Out[•]= {84, 822, 8211, 81111, 444, 4422, 44211, 441111,
 42222, 422211, 4221111, 42111111, 411111111, 222222, 2222211,
 22221111, 222111111, 2211111111, 21111111111, 111111111111}
```

Then $p_{\mathcal{B}}(12) = 20$:

```
In[•]:= Length@bin12
Out[•]= 20
```

Here is the sequence (A018819):

```
In[•]:= Table[Length@IntegerPartitions[n, All, 2^Range[0, 5]], {n, 20}]
Out[•]= {1, 2, 2, 4, 4, 6, 6, 10, 10, 14, 14, 20, 20, 26, 26, 36, 36, 46, 46, 60}
```

Euler’s generating function for $p_A(n)$

Consider the finite product of r infinite geometric progressions
$(1 + q^{a_1} + q^{2 a_1} + q^{3 a_1} + \ldots)(1 + q^{a_2} + q^{2 a_2} + q^{3 a_2} + \ldots)\ldots(1 + q^{a_r} + q^{2 a_r} + q^{3 a_r} + \ldots)$, where
$A = \{a_1, a_2, a_3, \ldots, a_r\}$ with $a_1 \geq a_2 \geq a_3 \geq \ldots \geq a_r$.

Multiplying out gives the general exponent of q to be

$f_1 a_1 + f_2 a_2 + \ldots + f_j a_j + \ldots,$

which is an arbitrary partition into elements of A where the term a_j is repeated f_j times.

Collecting terms with the same exponent n gives $p_A(n)$ terms q^n, so that the generating function for p_A is

$\sum_{n=0}^{\infty} p_A(n) = \prod_{a \in A}(1 + q^a + q^{2a} + q^{3a} + \ldots) = \prod_{a \in A} \frac{1}{1-q^a}$.

Riemann ζ function

This is similar to Euler’s product for the Riemann ζ function, which is based on the fundamental theorem of arithmetic.

$\sum_{n=0}^{\infty} \frac{1}{n^s} = \prod_{p\text{ prime}}(1 + p^{-s} + p^{-2s} + p^{-3s} + \ldots) = \prod_{p\text{ prime}} \frac{1}{1-p^{-s}}$.

The typical term consists of a product of a choice of finitely many terms, one from each series, like
${p_{j_1}}^{-f_1 s} {p_{j_2}}^{-f_2 s} \ldots {p_{j_r}}^{-f_r s}$, which is n^{-s} for that positive integer n with unique prime factorization
${p_{j_1}}^{f_1} {p_{j_2}}^{f_2} \ldots {p_{j_r}}^{f_r}$.

$p_A^{(s)}(n)$

Let $p_A^{(s)}(n)$ be the number of partitions of n into elements of A, where each part occurs at most s times.

11.2 Partition Analysis

Start with the multidimensional sum:

$\sum_{n=0}^{\infty} p_m(n)\, q^n = \sum_{n_1 \geq n_2 \geq \ldots \geq n_m} q^{n_1+n_2+\ldots+n_m}$

Consider the sum:

$\sum_{n_1,n_2,\ldots,n_m} q^{n_1+n_2+\ldots+n_m}\, \lambda_1{}^{n_1-n_2}\, \lambda_2{}^{n_2-n_3} \ldots \lambda_{m-1}{}^{n_{m-1}-n_m}$

“If we select only terms in the sum with nonnegative exponents on the λ, then the corresponding exponent will be a partition of n into $\leq m$ parts. For example, when $m = 2$ and $n = 4$ the exponents that result are 4+0, 3+1, and 2+2.” [Quotations are from chapter 11 of Special Functions.]

Here 5 plays the role of infinity, being greater than 2 and 4.

In[∘]:=
```
L = Sum[q^(n1+n2) λ1^(n1-n2), {n1, 0, 5}, {n2, 0, 5}]
```

Out[∘]=
$$1 + q^2 + q^4 + q^6 + q^8 + q^{10} + \frac{q^5}{\lambda_1^5} + \frac{q^4}{\lambda_1^4} + \frac{q^6}{\lambda_1^4} + \frac{q^3}{\lambda_1^3} + \frac{q^5}{\lambda_1^3} + \frac{q^7}{\lambda_1^3} + \frac{q^2}{\lambda_1^2} + \frac{q^4}{\lambda_1^2} + \frac{q^6}{\lambda_1^2} + \frac{q^8}{\lambda_1^2} + \frac{q}{\lambda_1} + \frac{q^3}{\lambda_1} + \frac{q^5}{\lambda_1} + \frac{q^7}{\lambda_1} + \frac{q^9}{\lambda_1} + q\,\lambda_1 + q^3\,\lambda_1 + q^5\,\lambda_1 + q^7\,\lambda_1 + q^9\,\lambda_1 + q^2\,\lambda_1^2 + q^4\,\lambda_1^2 + q^6\,\lambda_1^2 + q^8\,\lambda_1^2 + q^3\,\lambda_1^3 + q^5\,\lambda_1^3 + q^7\,\lambda_1^3 + q^4\,\lambda_1^4 + q^6\,\lambda_1^4 + q^5\,\lambda_1^5$$

“The method of partition analysis applies a linear operator $\Omega_{\geq}$ to such multiple Laurent series in $\lambda_1, \ldots, \lambda_{m-1}$. The operator annihilates terms with any negative exponents and in the remaining terms sets each $\lambda_i = 1$.”

In[∘]:=
```
S = L /. λ1^p_. → If[p < 0, 0, 1]
```

Out[∘]=
$$1 + q + 2\,q^2 + 2\,q^3 + 3\,q^4 + 3\,q^5 + 3\,q^6 + 2\,q^7 + 2\,q^8 + q^9 + q^{10}$$

Indeed $p_2(4) = 3$.

In[∘]:=
```
Coefficient[S, q^4]
```

Out[∘]= 3

“Hence

$\sum_{n=0}^{\infty} p_m(n)\, q^n =$

$\Omega_{\geq} \sum_{n_1,n_2,\ldots,n_m} q^{n_1+n_2+\ldots+n_m}\, \lambda_1{}^{n_1-n_2}\, \lambda_2{}^{n_2-n_3} \ldots \lambda_{m-1}{}^{n_{m-1}-n_m} =$

$\Omega_{\geq} \sum_{n_1 \geq 0} (q\,\lambda_1)^{n_1} \sum_{n_2 \geq 0} (q\,\lambda_2/\lambda_1)^{n_2} \ldots \sum_{n_m \geq 0} (q\,\lambda_{m-1})^{n_m} =$

$\Omega_{\geq} \frac{1}{(1-q\,\lambda_1)\,(1-q\,\lambda_2/\lambda_1) \ldots (1-q\,/\lambda_{m-1})}$. (11.2.1)”

Lemma 11.2.1

This begins to deal with the terms of (11.2.1):

$\Omega_{\geq} \frac{1}{(1-\lambda\,x)\,(1-y/\lambda)} = \frac{1}{(1-x)\,(1-x\,y)}$.

Theorem 11.2.2

Repeating the lemma boils away the λ’s from (11.2.1):

$\sum_{n=0}^{\infty} p_m(n)\, q^n = \frac{1}{(1-q)\,(1-q^2) \ldots (1-q^m)}$.

Lemma 11.2.3

If α is a nonnegative integer,

$$\Omega_{\geq} \frac{\lambda^{-\alpha}}{(1-\lambda x)(1-y/\lambda)} = \frac{x^{\alpha}}{(1-x)(1-x y)}.$$

The proof is left as an exercise, but instead of applying Lemma 11.2.1 as suggested, I imitated its proof; see the last section.

11.3 A Library for the Partition Analysis Algorithm

Proposition 11.3.1

(a) $\Omega_{\geq} \frac{1}{(1-\lambda x)((1-y_1/\lambda)(1-y_2/\lambda)\ldots(1-y_j/\lambda)} = \frac{1}{(1-x)((1-x y_1)\ldots(1-x y_j)}.$

(b) $\Omega_{\geq} \frac{1}{(1-\lambda x)(1-\lambda y)(1-z/\lambda)} = \frac{1-x y z}{(1-x)(1-y)(1-x z)(1-y z)}.$

(c) $\Omega_{\geq} \frac{1}{(1-\lambda x)(1-\lambda y)\left(1-z/\lambda^2\right)} = \frac{1+x y z-x^2 y z-x^2 y z}{(1-x)(1-y)\left(1-x^2 y\right)\left(1-y^2 z\right)}.$

The proof of (a) uses partial fractions and induction.

The proof of (b) also uses partial fractions; this verifies the algebra of the last step:

```
In[∘]:= Together[x/((x - y) (1 - x) (1 - x z)) - y/((x - y) (1 - y) (1 - y z))]
```

```
Out[∘]= (1 - x y z)/((-1 + x) (-1 + y) (-1 + x z) (-1 + y z))
```

11.4 Generating Functions
and
11.5 Some Results on Partitions

The following three theorems are proved using Lemma 11.2.3.

$Q_m(n)$

Let $Q_m(n)$ denote the number of partitions of n into exactly m distinct parts.

Theorem 11.4.1

Then

$$\sum_{n=0}^{\infty} Q_m(n)\, q^n = \frac{q^{m(m+1)/2}}{(1-q)\left(1-q^2\right)\ldots\left(1-q^m\right)}.$$

$Q_m^{(k,l)}(n)$

Let $Q_m^{(k,l)}(n)$ denote the number of partitions of n into m distinct parts where each part differs from the next by at least k and the smallest part is at least l.

Theorem 11.4.2

Then

$\sum_{n=0}^{\infty} Q_m^{(k,l)}(n)\, q^n = \frac{q^{l\,m+k\,m(m-1)/2}}{(1-q)\,(1-q^2)\ldots(1-q^m)}$.

$p_m(j, n)$

Suppose $p_m(j, n)$ denotes the number of partitions of n into at most m parts with largest part j.

$Q_m(j, n)$

Similarly, suppose $Q_m(j, n)$ denotes the number of partitions of n into *exactly m distinct* parts with largest part j.

Theorem 11.4.3

Then

$\sum_{n=0}^{\infty} p_m(j, n)\, z^j\, q^n = \frac{1}{(1-z\,q)\,(1-z\,q^2)\ldots(1-z\,q^m)}$

and

$\sum_{n=0}^{\infty} Q_m(j, n)\, z^j\, q^n = \frac{z^m\, q^{m(m+1)/2}}{(1-z\,q)\,(1-z\,q^2)\ldots(1-z\,q^m)}$.

$[z^j]$ notation

The operator $[z^j]$ applied to a power series in z gives the coefficient of z^j.

$p(N, M, n)$

Suppose $p(N, M, n)$ denotes the number of partitions of n into at most M parts, each at most N.

$Q(N, M, n)$

Suppose $Q(N, M, n)$ denotes the number of partitions of n into *exactly M distinct* , each at most N.

(The word “distinct” is missing in the book.)

Theorem 11.4.4

Then

$\sum_{n=0}^{\infty} p(N, M, n)\, q^n = \begin{bmatrix} N+M \\ M \end{bmatrix}_q$

and

$\sum_{n=0}^{\infty} Q(N, M, n)\, q^n = q^{M(M+1)/2} \begin{bmatrix} N \\ M \end{bmatrix}_q.$

Here $\begin{bmatrix} n \\ m \end{bmatrix}_q$ is the q-binomial coefficient defined by (10.0.5).

Theorem 11.4.4 ⇒ Theorem 11.5.1

$p(N, M, n) = p(M, N, n)$.

Because $\begin{bmatrix} N+M \\ M \end{bmatrix}_q = (q; q)_{N+M} / (q; q)_M\, (q; q)_{N+M-M} = (q; q)_{M+N} / (q; q)_N\, (q; q)_{M+N-M} = \begin{bmatrix} M+N \\ N \end{bmatrix}_q$.

For example with $n = 9$, $M = 5$, and $N = 3$,

```
In[•]:= Select[IntegerPartitions[9, 5], And @@ (# ≤ 3 & /@ #) &]
Out[•]= {{3, 3, 3}, {3, 3, 2, 1}, {3, 3, 1, 1, 1}, {3, 2, 2, 2}, {3, 2, 2, 1, 1}, {2, 2, 2, 2, 1}}
```

and switching N and M,

```
In[•]:= Select[IntegerPartitions[9, 3], And @@ (# ≤ 5 & /@ #) &]
Out[•]= {{5, 4}, {5, 3, 1}, {5, 2, 2}, {4, 4, 1}, {4, 3, 2}, {3, 3, 3}}
```

We get 6 in both cases.

⇒ Corollary 11.5.2

The number of partitions of n into at most m parts = the number of partitions of n into parts each at most m.

Put $m = M$ and $N = n$ in the theorem

```
In[•]:= Row /@ IntegerPartitions[9, 3]
Out[•]= {9, 81, 72, 711, 63, 621, 54, 531, 522, 441, 432, 333}
```

```
In[•]:= Length@%
Out[•]= 12
```

and then $M = n$ in the theorem.

```
In[•]:= Row /@ Select[IntegerPartitions[9], And @@ (# ≤ 3 & /@ #) &]
Out[•]= {333, 3321, 33111, 3222, 32211, 321111,
         3111111, 22221, 222111, 2211111, 21111111, 111111111}
```

```
In[•]:= Length@%
Out[•]= 12
```

Corollary 11.4.5

$\sum_{n,M\geq 0} Q(N, M, n)\, z^M q^n = (1 + z q) \ldots (1 + z q^N)$.

$p(n)$

Let $p(n)$ denote the total number of partitions of n.

$p(m, n)$

Let $p(m, n)$ denote the total number of partitions of n into exactly m parts.

Theorem 11.4.6

Then

(a) $\sum_{n=0}^{\infty} p(n)\, q^n = 1/(q; q)_\infty$,

(b) $\sum_{n=0}^{\infty} p(m, n)\, z^m q^n = 1/(z q; q)_\infty$,

(c) $\sum_{n=0}^{\infty} Q_m(n)\, z^m q^n = 1/(-z q; q)_\infty$,

(d) $\sum_{n=0}^{\infty} Q_m^{(2,1)}(n)\, z^m q^n = \sum_{m=0}^{\infty} z^m q^{m^2} / (q; q)_m$,

(e) $\sum_{n=0}^{\infty} Q_m^{(2,2)}(n)\, z^m q^n = \sum_{m=0}^{\infty} z^m q^{m(m+1)/2} / (q; q)_m$,

Partition analysis is applied to (b) and (c) to give other proofs of (a) and (b) of Corollary 10.2.2.

The series on the right of (d) and (e) occur in the Rogers–Ramanujan formulas, which will be stated and proved in the next chapter. These formulas can be interpreted in terms of partitions.

A distinct partition has gaps between each pair of parts at least 1. Gap-2 partitions have each pair of parts at least 2 apart.

```
In[∘]:= Gap2@x_ := Select[IntegerPartitions[x], -2 ≥ Max@Differences@# &]
```

For $n = 9$:

```
In[∘]:= Row /@ Gap2[9]
Out[∘]= {9, 81, 72, 63, 531}
```

Theorem 11.5.3, the Rogers–Ramanujan Identities

The number of gap-2 partitions of n

=

the number of partitions of n into parts congruent to 1 or 4 modulo 5.

The number of gap-2 partitions of n with parts at least 2

=

the number of partitions of n into parts congruent to 2 or 3 modulo 5.

For example, for $n = 12$,

Gap-2:

```
In[•]:= Row /@ Gap2[12]
Out[•]= {12, 111, 102, 93, 84, 831, 75, 741, 642}
```

```
In[•]:= Length@%
Out[•]= 9
```

Parts are 1 or 4 (mod 5).

```
In[•]:= Row /@ Select[IntegerPartitions[12], {} == Complement[Mod[#, 5], {1, 4}] &]
Out[•]= {111, 9111, 66, 6411, 6111111, 444, 441111, 411111111, 111111111111}
```

```
In[•]:= Length@%
Out[•]= 9
```

Gap-2 and each part at least 2.

```
In[•]:= Row /@ Select[Gap2[12], Min@# ≥ 2 &]
Out[•]= {12, 102, 93, 84, 75, 642}
```

```
In[•]:= Length@%
Out[•]= 6
```

Parts are 2 or 3 (mod 5).

```
In[•]:= Row /@ Select[IntegerPartitions[12], {} == Complement[Mod[#, 5], {2, 3}] &]
Out[•]= {12, 822, 732, 3333, 33222, 222222}
```

```
In[•]:= Length@%
Out[•]= 6
```

11.6 Graphical Methods

Ferrers Diagram

A Ferrers diagram shows each part of a partition as a row of dots of that length.

```
In[•]:= Ferrers@p_ := Framed@Grid[Table["●", #] & /@ p]
```

```
In[•]:= Ferrers@{5, 5, 1, 1, 1}
```

Out[•]=



Conjugate Partition

The conjugate partition is the partition formed by transposing the Ferrers diagram.

```
In[∘]:= ConjugatePartition[l_List] := Module[{i, r = Reverse[l], n = Length[l]},
          Table[n + 1 - Position[r, _? (# ≥ i &), Infinity, 1][[1, 1]], {i, l[[1]]}]]
```

```
In[∘]:= ConjugatePartition@{5, 5, 1, 1, 1}
Out[∘]= {5, 2, 2, 2, 2}
```

```
In[∘]:= Ferrers@%
```

Out[∘]=


This shows that $p(m, n)$ = the number of partitions of n with maximum part m. Similarly Theorem 11.5.1 and Corollary 11.5.2 follow.

Durfee Square

The Durfee square of a partition is the largest square in the Ferrers diagram of the partition.

```
In[∘]:= Flatten[Table[{k, m}, {k, 5}, {m, 2}], {2}]
Out[∘]= {{{1, 1}, {2, 1}, {3, 1}, {4, 1}, {5, 1}}, {{1, 2}, {2, 2}, {3, 2}, {4, 2}, {5, 2}}}
```

```
In[∘]:= Flatten[Outer[List, Range@5, Range@2], 1]
Out[∘]= {{1, 1}, {1, 2}, {2, 1}, {2, 2}, {3, 1}, {3, 2}, {4, 1}, {4, 2}, {5, 1}, {5, 2}}
```

Out[∘]=


```
In[∘]:= DurfeeSquare[s_List] :=
         Module[{i, max = 1}, Do[If[s⟦i⟧ ≥ i, max = i], {i, 2, Min[Length[s], First[s]]}];
          max]
```

This gives the size of the Durfee square.

```
In[∘]:= DurfeeSquare@{5, 5, 1, 1, 1}
Out[∘]= 2
```

Frobenius Symbol

Draw the diagonal of the Durfee square and make parts from the number of dots in the rows/columns to the right/below the diagonal.

Out[∘]=



In the figure the original partition is {5, 5, 1, 1, 1} and its Frobenius symbol is $\begin{pmatrix} 4 & 3 \\ 4 & 0 \end{pmatrix}$.

In[∘]:=

```
FrobeniusSymbol@x_ := Module[
  {d = DurfeeSquare@x},
  {Take[x, d] - Range@d, Take[ConjugatePartition@x, d] - Range@d}
 ]
```

In[∘]:=

```
FrobeniusSymbol@{5, 5, 1, 1, 1} // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 4 & 3 \\ 4 & 0 \end{pmatrix}$$

The two gray dots on the diagonal of the Durfee square *D* are not part of the symbol. The size of *D* is the number of columns in the symbol and so the original partition can be reconstructed.

In[∘]:=

```
PartitionFromSymbol@{a_, b_} := With[{d = Length@a},
  ReverseSort@Join[
    a + Range@d,
    ConjugatePartition[b - d + Range@d]
   ]
 ]
```

In[∘]:=

```
PartitionFromSymbol@{{4, 3}, {4, 0}}
```

Out[∘]=

```
{5, 5, 1, 1, 1}
```

Clearly the Frobenius symbol of the conjugate of a partition λ switches the rows of the Frobenius symbol of λ.

In[∘]:=

```
FrobeniusSymbol@ConjugatePartition@{5, 5, 1, 1, 1} // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 4 & 0 \\ 4 & 3 \end{pmatrix}$$

Applications

These graphical methods are used to prove three previous theorems.

Theorem 11.4.1 (p. 559):

$$\sum_{n=0}^{\infty} Q_m(n)\, q^n = \frac{q^{m(m+1)/2}}{(1-q)\left(1-q^2\right)\ldots\left(1-q^m\right)}.$$

The case $x = q$ in (10.9.2) of Corollary 10.9.4 (not 10.10.4 as stated in the text), p. 523:

$$1 + \sum_{n=1}^{\infty} \frac{q^{n^2}}{(1-q)^2 (1-q^2)^2 \ldots (1-q^m)^2} = \prod_{n=1}^{\infty} (1 - q^n)^{-1}.$$

The triple product identity (p. 496):

$$\frac{1}{(q;q)_\infty} \sum_{n=1}^{\infty} q^{\binom{n+1}{2}} x^n = (-x\, q;\, q)_\infty \, (-x^{-1};\, q)_\infty.$$

11.7 Congruence Properties of Partitions

Ramanujan’s congruences

Ramanujan discovered

$p(5\, n + 4) \equiv 0 \pmod 5$,

$p(7\, n + 5) \equiv 0 \pmod 7$,

based on a table constructed by MacMahon in about a month similar to this one:

In[•]:=
```
Grid[Partition[Table[PartitionsP@n, {n, 200}], 5], Alignment → Right]
```

Out[•]=

1	2	3	5	7
11	15	22	30	42
56	77	101	135	176
231	297	385	490	627
792	1002	1255	1575	1958
2436	3010	3718	4565	5604
6842	8349	10143	12310	14883
17977	21637	26015	31185	37338
44583	53174	63261	75175	89134
105558	124754	147273	173525	204226
239943	281589	329931	386155	451276
526823	614154	715220	831820	966467
1121505	1300156	1505499	1741630	2012558
2323520	2679689	3087735	3554345	4087968
4697205	5392783	6185689	7089500	8118264
9289091	10619863	12132164	13848650	15796476
18004327	20506255	23338469	26543660	30167357
34262962	38887673	44108109	49995925	56634173
64112359	72533807	82010177	92669720	104651419
118114304	133230930	150198136	169229875	190569292
214481126	241265379	271248950	304801365	342325709
384276336	431149389	483502844	541946240	607163746
679903203	761002156	851376628	952050665	1064144451
1188908248	1327710076	1482074143	1653668665	1844349560
2056148051	2291320912	2552338241	2841940500	3163127352
3519222692	3913864295	4351078600	4835271870	5371315400
5964539504	6620830889	7346629512	8149040695	9035836076
10015581680	11097645016	12292341831	13610949895	15065878135
16670689208	18440293320	20390982757	22540654445	24908858009
27517052599	30388671978	33549419497	37027355200	40853235313
45060624582	49686288421	54770336324	60356673280	66493182097
73232243759	80630964769	88751778802	97662728555	107438159466
118159068427	129913904637	142798995930	156919475295	172389800255
189334822579	207890420102	228204732751	250438925115	274768617130
301384802048	330495499613	362326859895	397125074750	435157697830
476715857290	522115831195	571701605655	625846753120	684957390936
749474411781	819876908323	896684817527	980462880430	1071823774337
1171432692373	1280011042268	1398341745571	1527273599625	1667727404093
1820701100652	1987276856363	2168627105469	2366022741845	2580840212973
2814570987591	3068829878530	3345365983698	3646072432125	3972999029388

Let us use Mathematica to verify Ramanujan’s generating function that proves the first congruence:

$$\sum_{n=0}^{\infty} p(5n+4)\,q^n = 5\,\frac{\{(1-q^5)(1-q^{10})(1-q^{15})\ldots\}^5}{\{(1-q)(1-q^2)(1-q^3)\ldots\}^6}$$

In[•]:=
```
Series[5 ((1 - q^5) (1 - q^10))^5 /
   ((1 - q) (1 - q^2) (1 - q^3) (1 - q^4) (1 - q^5) (1 - q^6) (1 - q^7) (1 - q^8) (1 - q^9) (1 - q^10))^6, {q,
  0, 10}]
```

Out[•]=

$$5 + 30\,q + 135\,q^2 + 490\,q^3 + 1575\,q^4 + 4565\,q^5 + 12310\,q^6 + 31185\,q^7 + 75175\,q^8 + 173525\,q^9 + 386155\,q^{10} + O[q]^{11}$$

Indeed,

In[∘]:=
```
Table[PartitionsP[5 n + 4], {n, 0, 10}]
```
Out[∘]=
```
{5, 30, 135, 490, 1575, 4565, 12310, 31185, 75175, 173525, 386155}
```

Solutions

Proof of Lemma 11.2.3

“If α is a nonnegative integer,

$\Omega_{\geq} \frac{\lambda^{-\alpha}}{(1-\lambda x)(1-y/\lambda)} = \frac{x^\alpha}{(1-x)(1-x y)}$.”

Retracing the proof of Lemma 11.2.1 as suggested, the left side equals

$\Omega_{\geq} \sum_{n,m\geq 0} \lambda^{-\alpha+n-m} x^n y^m = \sum_{n\geq m\geq \alpha} x^n y^m$.

Set $k = n - m - \alpha$ so that the last sum becomes

$\sum_{k,m\geq 0} x^{m+k+\alpha} y^m = x^\alpha \sum_{k\geq 0} x^k \sum_{m\geq 0} (x y)^m = \frac{x^\alpha}{(1-x)(1-x y)}$.

Proof of Lemma 11.2.3 (c)

The lemma states:

$\frac{1}{(1-\lambda x)(1-\lambda y)(1-z/\lambda^2)} = \frac{1+x y z-x^2 y z-x^2 y z}{(1-x)(1-y)(1-x^2 y)(1-y^2 z)}$.

Following the proof of Lemma 11.2.1, call the following lemma C:

$\Omega_{\geq} \frac{1}{(1-x \lambda)(1-z/\lambda^2)} = \frac{1}{(1-x)(1-x^2 z)}$.

Proof

$\Omega_{\geq} \frac{1}{(1-x \lambda)(1-z/\lambda^2)} = \Omega_{\geq} \sum_{n,m\geq 0} \lambda^{n-2m} x^n z^m = \sum_{n\geq 2m\geq 0} x^n z^m$.

Set $k = n - 2m$ so that the last sum becomes

$\sum_{k,m\geq 0} x^{2m+k} z^m = \sum_{k\geq 0} x^k \sum_{m\geq 0} (x^2 z)^m = \frac{1}{(1-x)(1-x^2 z)}$. □

Proof of Lemma 11.2.3 (c):

The left-hand side of (c) is

$\Omega_{\geq} \frac{1}{(1-\lambda x)(1-\lambda y)(1-z/\lambda^2)} = \frac{1}{x-y} \Omega_{\geq}\left(\frac{x}{1-x\lambda} - \frac{y}{1-y\lambda}\right) \frac{1}{1-z/\lambda^2} = \frac{1}{x-y}\left(x\, \Omega_{\geq} \frac{1}{(1-x\lambda)(1-z/\lambda^2)} - y\, \Omega_{\geq} \frac{1}{(1-y\lambda)(1-z/\lambda^2)}\right)$.

Apply lemma C to each inside term (with x replaced by y in the second one) to get

$\frac{1}{x-y}\left(\frac{x}{(1-x)(1-x^2 z)} - \frac{y}{(1-y)(1-y^2 z)}\right)$

Use a machine to add the fractions (and switch the signs of the four terms in the denominator):

In[∘]:= Together$\left[\frac{1}{x-y}\left(\frac{x}{(1-x)\left(1-x^2 z\right)}-\frac{y}{(1-y)\left(1-y^2 z\right)}\right)\right]$

Out[∘]= $\frac{1+x y z-x^2 y z-x y^2 z}{(-1+x)(-1+y)\left(-1+x^2 z\right)\left(-1+y^2 z\right)}$